<commit_before>/* Based on http://blog.phil-opp.com/rust-os/multiboot-kernel.html
 *
 * Used to specify a custom linking layout that puts our multiboot header
 * before everything else.
 */

OUTPUT_FORMAT("elf64-x86-64")
OUTPUT_ARCH("i386:x86-64")
ENTRY(start)

SECTIONS {
    /* Load the kernel reasonably high in memory to avoid special addresses. */
    . = 1M;

	kernel_start = .;
    .boot :
    {
        /* This goes first. */
        KEEP(*(.multiboot_header))
    }

    .text ALIGN(4096) : AT(ADDR(.text))
    {
        *(.text)
		*(.text.*)
    }

	.rodata : AT(ADDR(.rodata))
	{
		*(.rodata)
		*(.rodata.*)
	}

	.data ALIGN(4096) : AT(ADDR(.data))
    {
        *(.data)
		*(.data.*)
    }

	.bss ALIGN(4096) : AT(ADDR(.bss))
    {
		__bss_start = .;
        *(.bss)
		*(.bss.*)
    }
	kernel_end = .;
}<commit_msg>use tabs instead of white spaces<commit_after>/* Based on http://blog.phil-opp.com/rust-os/multiboot-kernel.html
 *
 * Used to specify a custom linking layout that puts our multiboot header
 * before everything else.
 */

OUTPUT_FORMAT("elf64-x86-64")
OUTPUT_ARCH("i386:x86-64")
ENTRY(start)

SECTIONS {
	/* Load the kernel reasonably high in memory to avoid special addresses. */
	. = 1M;

	kernel_start = .;
	.boot :
	{
		/* This goes first. */
		KEEP(*(.multiboot_header))
	}

	.text ALIGN(4096) : AT(ADDR(.text))
	{
		*(.text)
		*(.text.*)
	}

	.rodata : AT(ADDR(.rodata))
	{
		*(.rodata)
		*(.rodata.*)
	}

	.data ALIGN(4096) : AT(ADDR(.data))
	{
		*(.data)
		*(.data.*)
	}

	.bss ALIGN(4096) : AT(ADDR(.bss))
	{
		__bss_start = .;
		*(.bss)
		*(.bss.*)
	}
	kernel_end = .;
}
{'fraction_non_alphanumeric': 0.18917237087741132, 'fraction_numerical': 0.03235843186060983, 'mean_word_length': 2.4956521739130433, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5363900', 'n_tokens_mistral': 715, 'n_tokens_neox': 654, 'n_words': 129}